
A SURFACE OF DEGREE 24 WITH 1440 SINGULARITIES OF TYPE D4

by

CÉDRIC BONNAFÉ

Abstract. — Using the invariant algebra of the reflection group denoted by G32 in Shephard-
Todd classification, we construct three irreducible surfaces in P3(C) with many singularities:
one of them has degree 24 and contains 1440 quotient singularities of type D4.

Let µD4
(d ) denote the maximal number of quotient singularities of type D4 that an

irreducible projective surface in P3(C)might have. Miyaoka [Miy] proved that

µD4
(d ) ¶

16

117
d (d −1)2

For d = 8, 16 or 24, this reads

µD4
(8) ¶ 53, µD4

(16) ¶ 492 and µD4
(24) ¶ 1736.

The main results of this note are that

(1) µD4
(8) ¾ 44, µD4

(16) ¾ 352 and µD4
(24) ¾ 1440

and that

(2) µD4
(8k ) ¾ 44k 3

for all k ¾ 1. So, let C[x1, x2, x3, x4] be the polynomial ring over C in 4 indeterminates with
its usual grading and let P3(C) = Proj(C[x1, x2, x3, x4]) be the projective space of dimension
3. If f ∈ C[x1, x2, x3, x4] is homogeneous, we denote by Z ( f ) the projective hypersurface
it defines. If k is a natural number, we denote by f [k ] the homogeneous polynomial
f (x k

1 , x k
2 , x k

3 , x k
4 ). We aim to prove the following result:

Theorem. There exists a polynomial g of degree 8 such that:
(a) Z (g ) is an irreducible surface of degree 8 in P3(C) with exactly 44 singular points which are

all quotient singularities of type D4.
(b) If k ¾ 1, then Z (g [k ]) is an irreducible surface of degree 8k with at least 44k 3 quotient

singularities of type D4.
(c) Z (g [2]) is an irreducible surface of degree 16 with exactly 472 singular points: 24 quotient

singularities of type A1, 96 quotient singularities of type A2 and 352 quotient singularities
of type D4.

(d) Z (g [3]) is an irreducible surface of degree 24 in P3(C) with exactly 1440 singular points
which are all quotient singularities of type D4. The automorphism group ofZ (g [3]) contains
at least 25920 elements and acts transitively on the 1440 singular points.

The author is partly supported by the ANR (Project No ANR-16-CE40-0010-01 GeRepMod).

The proof relies on extensive MAGMA computations [Magma], and goes as follow. Let

This is the polynomial of the Theorem. It has coefficients in Q but we will check in our
computations that the singular points of Z (g ) and Z (g [3]) have coordinates in the cyclo-
tomic field K =Q(z ), where z is a primitive 12-th root of unity. We denote by Z (g )sing the
(reduced) singular locus of Z (g ) respectively.

Remark 1. As we will see in the computations below, the polynomial g [3] is invariant
under the action of the complex reflection group denoted by G32 in Shephard-Todd clas-
sification. In fact, we did not find g directly: we found first g [3] by looking at invariants
of degree 24 of G32, following ideas of Barth [Bar] (for constructing Barth sextic with 65
nodes) and Sarti [Sar1], [Sar2], [Sar3] (for constructing for instance the Sarti surface of
degree 12 with 600 nodes), who used invariants of Coxeter groups of type H3 and H4. �

Remark 2. As the expert in computer algebra will notice by reading the MAGMA codes
below, the author is not a specialist in programming and asks for some indulgence. �

So let us start by defining the polynomial g in MAGMA. We will define it over the
rational numbers. This will not be important for the computations regarding Z (g ), but
this will shorten drastically computations regarding Z (g [3]).

> Q:=RationalField();
> P3<x1,x2,x3,x4>:=ProjectiveSpace(Q,3);
>
> g:=x1^8 - 6*x1^6*x2^2 - 60*x1^6*x2*x3
> - 60*x1^6*x2*x4 - 6*x1^6*x3^2 + 60*x1^6*x3*x4
> - 6*x1^6*x4^2 + 2240*x1^5*x2^2*x3
> - 2240*x1^5*x2^2*x4 + 2240*x1^5*x2*x3^2
> - 2240*x1^5*x2*x4^2 + 2240*x1^5*x3^2*x4
> - 2240*x1^5*x3*x4^2 - 14*x1^4*x2^4
> + 10180*x1^4*x2^3*x3 + 10180*x1^4*x2^3*x4
> + 40412*x1^4*x2^2*x3^2 - 23440*x1^4*x2^2*x3*x4
> + 40412*x1^4*x2^2*x4^2 + 10180*x1^4*x2*x3^3
> + 23440*x1^4*x2*x3^2*x4 + 23440*x1^4*x2*x3*x4^2
> + 10180*x1^4*x2*x4^3 - 14*x1^4*x3^4
> - 10180*x1^4*x3^3*x4 + 40412*x1^4*x3^2*x4^2
> - 10180*x1^4*x3*x4^3 - 14*x1^4*x4^4
> + 10180*x1^3*x2^4*x3 - 10180*x1^3*x2^4*x4
> + 111980*x1^3*x2^3*x3^2 - 111980*x1^3*x2^3*x4^2
> + 111980*x1^3*x2^2*x3^3 - 111980*x1^3*x2^2*x4^3
> + 10180*x1^3*x2*x3^4 - 10180*x1^3*x2*x4^4
> + 10180*x1^3*x3^4*x4 - 111980*x1^3*x3^3*x4^2
> + 111980*x1^3*x3^2*x4^3 - 10180*x1^3*x3*x4^4
> - 6*x1^2*x2^6 + 2240*x1^2*x2^5*x3 + 2240*x1^2*x2^5*x4
> + 40412*x1^2*x2^4*x3^2 + 23440*x1^2*x2^4*x3*x4
> + 40412*x1^2*x2^4*x4^2 + 111980*x1^2*x2^3*x3^3
> + 111980*x1^2*x2^3*x4^3 + 40412*x1^2*x2^2*x3^4
> + 154704*x1^2*x2^2*x3^2*x4^2 + 40412*x1^2*x2^2*x4^4
> + 2240*x1^2*x2*x3^5 - 23440*x1^2*x2*x3^4*x4
> - 23440*x1^2*x2*x3*x4^4 + 2240*x1^2*x2*x4^5
> - 6*x1^2*x3^6 - 2240*x1^2*x3^5*x4
> + 40412*x1^2*x3^4*x4^2 - 111980*x1^2*x3^3*x4^3
> + 40412*x1^2*x3^2*x4^4 - 2240*x1^2*x3*x4^5
> - 6*x1^2*x4^6 - 60*x1*x2^6*x3 + 60*x1*x2^6*x4
> + 2240*x1*x2^5*x3^2 - 2240*x1*x2^5*x4^2
> + 10180*x1*x2^4*x3^3 - 23440*x1*x2^4*x3^2*x4
> + 23440*x1*x2^4*x3*x4^2 - 10180*x1*x2^4*x4^3
> + 10180*x1*x2^3*x3^4 - 10180*x1*x2^3*x4^4
> + 2240*x1*x2^2*x3^5 + 23440*x1*x2^2*x3^4*x4
> - 23440*x1*x2^2*x3*x4^4 - 2240*x1*x2^2*x4^5
> - 60*x1*x2*x3^6 + 23440*x1*x2*x3^4*x4^2
> - 23440*x1*x2*x3^2*x4^4 + 60*x1*x2*x4^6 - 60*x1*x3^6*x4
> - 2240*x1*x3^5*x4^2 + 10180*x1*x3^4*x4^3
> - 10180*x1*x3^3*x4^4 + 2240*x1*x3^2*x4^5
> + 60*x1*x3*x4^6 + x2^8 - 6*x2^6*x3^2 - 60*x2^6*x3*x4
> - 6*x2^6*x4^2 - 2240*x2^5*x3^2*x4 - 2240*x2^5*x3*x4^2
> - 14*x2^4*x3^4 + 10180*x2^4*x3^3*x4
> + 40412*x2^4*x3^2*x4^2 + 10180*x2^4*x3*x4^3
> - 14*x2^4*x4^4 - 10180*x2^3*x3^4*x4
> - 111980*x2^3*x3^3*x4^2 - 111980*x2^3*x3^2*x4^3
> - 10180*x2^3*x3*x4^4 - 6*x2^2*x3^6 + 2240*x2^2*x3^5*x4
> + 40412*x2^2*x3^4*x4^2 + 111980*x2^2*x3^3*x4^3
> + 40412*x2^2*x3^2*x4^4 + 2240*x2^2*x3*x4^5 - 6*x2^2*x4^6
> + 60*x2*x3^6*x4 - 2240*x2*x3^5*x4^2 - 10180*x2*x3^4*x4^3
> - 10180*x2*x3^3*x4^4 - 2240*x2*x3^2*x4^5 + 60*x2*x3*x4^6
> + x3^8 - 6*x3^6*x4^2 - 14*x3^4*x4^4 - 6*x3^2*x4^6 + x4^8;

1. Degree 8

As explained above, degree 8 is sufficiently small so that all computations can be per-
formed directly over the field K .

> K<z>:=CyclotomicField(12);
> P3K<[x]>:=ProjectiveSpace(K,3);
> time Zg:=Surface(P3K,Evaluate(g,[x[1],x[2],x[3],x[4]]));
Time: 11.040
> Zgsing:=SingularSubscheme(Zg);
> Dimension(Zgsing);
0
> unionhyp:=Scheme(P3K,x[1]*x[2]*x[3]*x[4]);
> Dimension(Intersection(Zgsing,unionhyp));
-1

Note that the fact that dimZ (g )sing = 0 implies that Z (g ) is geometrically irreducible.
Note also that the last command implies that all the singular points are contained (for
instance) in the affine chart Z (g )aff defined by “x4 6= 0”. We will make all the remaining
computations in this affine chart:

> Zgaff:=AffinePatch(Zg,4);
> Zgaffsing:=SingularSubscheme(Zgaff);
> irrg:=IrreducibleComponents(Zgaffsing);
> irrg:=[ReducedSubscheme(i) : i in irrg];
> Set([Degree(i) : i in irrg]);
{ 1 }
> # irrg;
44

The last two commands shows that all singular points have coordinates in K and that
there are 44 singular points. We now check that they are quotient singularities of type D4:

> sings:=[RationalPoints(i) : i in irrg];
> sings:=&cat [[Coordinates(i) : i in j] : j in sings];
> sings:=[Zgaff ! i : i in sings];
> test:=[ ];
> for i in sings
for> do a,b,c:=IsSimpleSurfaceSingularity(i);
for> test:=test cat [[* a,b,c *]];
for> end for;
> test[1];
[* true, D, 4 *]
> Set([i eq test[1] : i in test]);
{ true }

This concludes the proof of statement (a) of the Theorem. Note that the surface Z (g ) has
only 8 real points: Figure 1 shows part of the real locus of Z (g ).

FIGURE 1. Part of the real locus of Z (g )

2. Degree 8k

Let U denote the open subset of P3(C) defined by x1 x2 x3 x4 6= 0 and let σk : P3(C) →
P3(C), [x1; x2; x3; x4] 7→ [x k

1 , x k
2 , x k

3 , x k
4 ]. The restriction of σk to a morphism U →U is an

étale Galois covering, with group (µk )
4/∆µk (here, µk denotes the group of k -th roots of

unity in C× and ∆ :µk ,→ (µk )
4 is the diagonal embedding). We have Z (g [k ]) =σ∗k (Z (g )).

Let us first prove that Z (g [k ]) is irreducible. We may assume that k ¾ 2, as the result
has been proved for k = 1 in the previous section. Note the following fact: if i 6= j , then
the scheme Zi , j defined by the ideal 〈g , ∂ g

∂ xi
, ∂ g
∂ x j
〉 has dimension 0:

> dg:=[Derivative(g,i) : i in [1..4]];
> Dimension(Scheme(P3,[g,dg[1],dg[2]]));
0
> Dimension(Scheme(P3,[g,dg[1],dg[3]]));
0
> Dimension(Scheme(P3,[g,dg[1],dg[4]]));
0
> Dimension(Scheme(P3,[g,dg[2],dg[3]]));
0
> Dimension(Scheme(P3,[g,dg[2],dg[4]]));
0
> Dimension(Scheme(P3,[g,dg[3],dg[4]]));
0

Recall that
∂ g [k ]
∂ xi

= k x k−1
i (

∂ g

∂ xi
◦σk ),

so the singular locus of Z (g [k ]) is contained in

{p1, p2, p3, p4}∪
�⋃

i 6= j

σ∗k (Zi , j

�

,

where pi = [δi 1;δi 2;δi 3;δi 4] (and δi j is the Kronecker symbol). Since σk is finite, this
implies that Z (g [k ])sing has dimension 0, so Z (g [k ]) is irreducible.

Now, as σk : U → U is étale and the singular locus of Z (g ) is contained in U (see
the command Dimension(Intersection(Zgsing,unionhyp)) of the previous
section), the 44 singularities of Z (g ) give 44k 3 singularities in Z (g [k ]) of the same type,
i.e. quotient singularities of type D4. This proves the statement (b) of the Theorem.

Note that, for k = 2, 3 and 4 (and maybe for bigger k ) we will prove in the next sections
that Z (g [k ]) contains singular points outside of U .

3. Degree 16

Using the morphism σ2 defined in the previous section and the fact that it is étale over
U , we get that Z (g [2])∩U has exactly 352 singular points, which are all quotient singu-
larities of type D4. We now need to determine the singularities which are not contained
in U . So let H be the complement of U in P3(C).

> g2:=Evaluate(g,[x1^2,x2^2,x3^2,x4^2]);
> Z2:=Surface(P3,g2);
> Z2sing:=SingularSubscheme(Z2);
> time Dimension(Z2sing);
0
Time: 20.920
> unionhyp:=Scheme(P3,x1*x2*x3*x4);
> Z2singhyp:=Intersection(Z2sing,unionhyp);
> Z2singhyp:=ReducedSubscheme(Z2singhyp);
> irrsinghyp:=IrreducibleComponents(Z2singhyp);
> # irrsinghyp;
18
> Set([Degree(i) : i in irrsinghyp]);
{ 4, 8 }
> &+ [Degree(i) : i in irrsinghyp];
120

This shows that Z (g [2])sing ∩H contains 120 points, none of which being defined over
Q. We denote by a an 8-th root of unity and by K4 the group the cyclotomic field Q(a ).
We denote by K8 the splitting field of the polynomial

T 16+3248T 12+23100000T 8+20300000000T 4+39062500000000,

and by u a root of this polynomial (one can check that K8 is a Galois extension of Q).
It turns out that every singular point which is defined over a field extension of degree 4

(resp. 8) has coordinates in K4 (resp. K8). We separate the treatment of these two families.

Singular points defined over K4. We will check that all of them are quotient singularities
of type A1:

> singhyp4:=[i : i in irrsinghyp | Degree(i) eq 4];
> K4<a>:=CyclotomicField(8);
> Z2K4:=ChangeRing(Z2,K4);
> P3K4<[x]>:=AmbientSpace(Z2K4);
> singhyp4:=[ChangeRing(i,K4) : i in singhyp4];
> singhyp4:=[* [j : j in RationalPoints(i)]
> : i in singhyp4 *];
> singhyp4:=[Coordinates(i[1]) : i in singhyp4];
> singhyp4:=[Z2K4 ! i : i in singhyp4];

> types:=[];
> for i in singhyp4 do
for> te,ty,ra:=IsSimpleSurfaceSingularity(i);
for> types:=types cat [[* te,ty,ra *]];
for> end for;
> types;
[ [* true, A, 1 *], [* true, A, 1 *],
[* true, A, 1 *], [* true, A, 1 *],
[* true, A, 1 *], [* true, A, 1 *] ]

We have chosen one point in each irreducible components over Q, so this proves that
Z (g ) contains at least 24 quotient singularities of tpe A1.

Singular points defined over K8. We will check that all of them are quotient singularities
of type A2. So let us first pick one point in each irreducible component over Q:

> singhyp8:=[i : i in irrsinghyp | Degree(i) eq 8];
> POL<T>:=PolynomialRing(Q);
> K8<u>:=NumberField(T^16 + 3248*T^12
> + 23100000*T^8 + 20300000000*T^4
> + 39062500000000);
> Z2K8:=ChangeRing(Z2,K8);
> P3K8<[x]>:=AmbientSpace(Z2K8);
> singhyp8:=[ChangeRing(i,K8) : i in singhyp8];
> singhyp8:=[* [j : j in RationalPoints(i)]
> : i in singhyp8 *];
> singhyp8:=[Coordinates(i[1]) : i in singhyp8];
> singhyp8:=[Z2K8 ! i : i in singhyp8];

The Magma command IsSimpleSurfaceSingularity does not answer in a rea-
sonable amount of time. Let us explain how we proceed for one of the points in the list
singhyp8. So let p be the first point in the list. It lives in the affine chart defined by
“x3 6= 0”, and we denote by (x , y , z ) the variables (x1/x3, x2/x3, x4/x3). Let f (x , y , z ) the
defining polynomial of Z (g [2]) in this affine chart, after translating p at the origin (0, 0, 0).
We denote by f2 its homogeneous component of degree 2:

> A3K8<x,y,z>:=AffineSpace(K8,3);
> p:=singhyp8[1];
> cop:=Coordinates(p);
> f:=Evaluate(g2,[x+cop[1],y+cop[2],1,z+cop[4]]);
> cof:=Coefficients(f);
> mof:=Monomials(f);
> l:=# mof;
> f2:=&+ [cof[i]*mof[i] : i in [1..l] |
> Degree(mof[i]) eq 2];
> f2:=f2/Reverse(Coefficients(f2))[1];
> Factorization(f2-z^2);
[

<x + 1/26400000000000*(-57*u^14 - 5500*u^12
- 372636*u^10 - 17864000*u^8
- 1456950000*u^6 - 92675000000*u^4
- 3075225000000*u^2 -

55825000000000)*y, 2>
]

FIGURE 2. Part of the real locus of Z (g [2])

The last command shows that, up to a change of coordinates, f2 might be transformed to
y 2 + z 2. By standard arguments, this proves that p is a quotient singularity of type Ak ,
for some k ¾ 2, which can be obtained as the Milnor number of Z (g [2]) at p : however,
Magma cannot compute this Milnor number in a reasonable amount of time and we need
to copy the polynomial f in the software Singular [DGPS] to compute this Milnor
number (!): we obtain 2. So p is a quotient singularity of type A2.

Repeating 12 times the same method (we omit details), we get that Z (g [2]) has 96
quotient singularities of type A2. This proves statement (c) of the Theorem.

Figure 2 shows part of the real locus of Z (g ).

4. Degree 24

We now come to the surface Z (g [3]). For shortening the first computations, we will
define it over Q.

> g3:=Evaluate(g,[x1^3,x2^3,x3^3,x4^3]);
> Zg3:=Surface(P3,g3);
> Zg3sing:=SingularSubscheme(Zg3);
> time Dimension(Zg3sing);
0
Time: 13.210

Again, the fact that the dimZ (g [3])sing = 0 implies thatZ (g [3]) is geometrically irreducible.
We will compute the singular locus of Z (g [3]) over Q, by separating its intersection with
the affine chart “x1 6= 0” and its intersection with the hyperplane “x1 = 0”.

> Zg3aff:=AffinePatch(Zg3,1);
> Zg3affsing:=SingularSubscheme(Zg3aff);
> time irrg3aff:=IrreducibleComponents(Zg3affsing);
Time: 48.370
> irrg3aff:=[ReducedSubscheme(i) : i in irrg3aff];
> hyp:=Intersection(Zg3sing,Scheme(P3,x1));
> irrhyp:=IrreducibleComponents(hyp);
> irrhyp:=[ReducedSubscheme(i) : i in irrhyp];

We now check that all the singular points are rational over the cyclotomic field K =Q(z ):

> K<z>:=CyclotomicField(12);
> irrg3aff:=[ChangeRing(i,K) : i in irrg3aff];
> irrg3aff:=&cat [IrreducibleComponents(i) : i in irrg3aff];
> irrhyp:=[ChangeRing(i,K) : i in irrhyp];
> irrhyp:=&cat [IrreducibleComponents(i) : i in irrhyp];
> Set([Degree(i) : i in irrg3aff]);
{ 1 }
> Set([Degree(i) : i in irrhyp]);
{ 1 }
> (# irrg3aff) + (# irrhyp);
1440

The above last command shows that Z (g [3]) has exactly 1440 singular points. We now
pick a singular point p in Z (g [3]), and show that it is a quotient singularity of type D4:

> Zg3K:=ChangeRing(Zg3,K);
> p:= Zg3K ! [0,-(2*z^3-z^2-z+1), (z^3+z^2+z), 2];
> IsSingular(Zg3K,p);
true
> IsSimpleSurfaceSingularity(p);
true D 4

We could prove that all other singular points are quotient singularities of type D4 (by
making 1440 computations!), but we prefer to do it using group theory. So let W be
the complex reflection group denoted by G32 in Shephard-Todd classification (see for in-
stance [LeTa]). We could obtain it by the MAGMA command W:=ShephardTodd(32),
but it turns out that the group implemented in this way in MAGMA does not stabilize
g [3]: with the version of G32 implemented in MAGMA, we would need to replace g [3] by
a much more complicated polynomial with huge coefficients lying in the cyclotomic field
Q(z 4). This would have lead to very long computations (we could even not check that the
1440 singular points are the only ones: MAGMA computations do not conclude, at least
after few hours). We prefer to use the model implemented by Jean Michel [Mic] in the
CHEVIE package of GAP3.

> a:=z^4;
> s1:=Matrix(K,4,4,
> [ [ 1, 0, 0, 0 ],
> [ 0, 1, 0, 0 ],
> [ 0, 0, a, 0 ],
> [ 0, 0, 0, 1 ] ]);
>
> s2:=Matrix(K,4,4,
> [ [-1/3*a-2/3*a^2, 2/3*a+1/3*a^2, 2/3*a+1/3*a^2, 0 ],
> [ 2/3*a+1/3*a^2,-1/3*a-2/3*a^2, 2/3*a+1/3*a^2, 0 ],
> [ 2/3*a+1/3*a^2, 2/3*a+1/3*a^2,-1/3*a-2/3*a^2, 0 ],
> [ 0, 0, 0, 1 ] ]);
>
> s3:=Matrix(K,4,4,
> [ [ 1, 0, 0, 0 ],
> [ 0, a, 0, 0 ],
> [ 0, 0, 1, 0 ],
> [ 0, 0, 0, 1 ] ]);
>
> s4:=Matrix(K,4,4,

> [ [-1/3*a-2/3*a^2,-2/3*a-1/3*a^2, 0,-2/3*a-1/3*a^2 ],
> [-2/3*a-1/3*a^2,-1/3*a-2/3*a^2, 0, 2/3*a+1/3*a^2 ],
> [ 0, 0, 1, 0 ],
> [-2/3*a-1/3*a^2, 2/3*a+1/3*a^2, 0,-1/3*a-2/3*a^2 ] ]);
>
> W:=MatrixGroup<4,K | [s1,s2,s3,s4]>;
> Order(W);
155520
> Order(W/Centre(W));
25920
> test:=IsIsomorphic(W,ShephardTodd(32));
> test;
true

We then check that g [3] is invariant under the action of W :

> R:=InvariantRing(W);
> P<X1,X2,X3,X4>:=PolynomialRing(R);
> g3copy:=Evaluate(g3,[X1,X2,X3,X4]);
> [g3copy^w eq g3copy : w in Generators(W)];
[ true, true, true, true ]

This shows that the group W defined as above, which is a complex reflection group
of type G32 in Shephard-Todd classification, acts on Z . Its center Z(W ) has order 6, so
W /Z(W ) has order 25920 and acts faithfully on Z . We conclude by computing the W -
orbit of the point p . We do this by first computing in the vector space V ' C4 and then
projecting on P(V )'P3(C):

> P3K:=ProjectiveSpace(K,3);
> ZW:=Centre(W);
> rep:=Transversal(W,ZW);
> V:=VectorSpace(W);
> v:= V ! Coordinates(p);
> orbv:=[v^(Transpose(w)) : w in rep];
> orbp:=[P3K ! Coordinates(V,i) : i in orbv];
> orbp:=Set(orbp);
> # orbp;
1440

As the W -orbit of p has 1440 elements and as Z (g [3]) has 1440 singular points, the proof
of statement (b) of the Theorem is complete. Note also that, in the given model, the
surface Z (g [3]) has only 32 real points: Figure 3 gives partial views of its real locus.

Remark 3. From Section 2, we deduce that Z (g [4])sing has 2816 quotient singularities of
type D4 in the open subset U and it can be checked that it has 480 other singular points
outside of U , for which we did not determine the type. �

Remark 4. After investigations in the invariant rings of several complex reflection groups,
we have also been able to construct curves with many singularities. For example:
• Using the reflection group W =G24, we have obtained a cuspidal curve of degree 14

in P2(C)with exactly 42 cusps (all lying in a single W -orbit). Note that W /Z(W ) has
order 168 and is isomorphic to GL3(F2).
• Using the reflection group W =G26, we have obtained a curve of degree 18 in P2(C)

with 72 cusps and 12 nodes (these are the two W -orbits of singular points). Note
that W /Z(W ) has order 216.

FIGURE 3. Part of the real locus of Z (g [3])

Also, other singular surfaces have been obtained. For example:
• Using the reflection group W = G29 (note that W /Z(W ) has order 1920), we have

obtained:
- a surface of degree 8 in P3(C) with 160 nodes, all belonging to the same W -

orbit. Recall that Endraß octic surface [End] has degree 8 and 168 nodes and its
automorphism group has order 16.

- a surface of degree 8 in P3(C) with 20 singular points of multiplicity 3 and
Milnor number 11, all belonging to the same W -orbit.

• Using the reflection group W = G31 (note that W /Z(W ) has order 11520), we have
obtained a surface of degree 20 in P3(C) with 1920 nodes, all lying in the same W -
orbit. Recall that Chmutov surface [Chm] of degree 20 has 2926 nodes.

Details will appear in a forthcoming paper. �

